# High-Tune: gas optical properties file format

November 28, 2018

**|Méso|Star>**
(https://www.meso-star.com)

## Contents

# 1 Context

This document describes the format of the file that provides all required optical properties (absorption and scattering coefficients, both in the LW and SW regions) of the **gas mixture** only. This data file uses the ASCII format, and the following convention: each data array is first described by a comment, then the numeric data is provided. A single numeric value is given for each line. A array of size $N$ is therefore provided over $N$ lines.

# 2 Description of the model and data usage

Data is provided for a stratified plane-parallel atmosphere: the atmosphere is described using $N_{lev}$ (pressure, or altitude) levels that define $N_{lay} = N_{lev} - 1$ layers. Some data is defined at each level, some is defined for layers (using the hypothesis of homogeneous layers). Finally, two spectral domains are described: the longwave (LW) and the shortwave (SW) domains. For each one of these domains, a number of spectral intervals are defined, and a number of quadrature points is given in each interval (variable).




Figure 1: Evolution of the absorption coefficient for the 10 quadrature points of the first LW spectral interval, for the first atmospheric layer (MLS profile) as a function of water vapor molar fraction.

Optical properties of the gas mixture (absorption and scattering coefficients) can be computed from the data provided in this file, and from the local water vapor content (provided in the LES data file). The atmospheric data file contains the values of the absorption and scattering coefficients of the gas mixture, tabulated as a function of the water vapor molar fraction $x_{H_2O}$. Figure 1 shows that the logarithm of the absorption coefficient evolves linearly with the logarithm of $x_{H_2O}$ over a wide range of water vapor molar fraction, for each quadrature point. Interpolating the optical properties of the gas mixture for any given value of $x_{H_2O}(local)$ is therefore very simple using the data of the atmospheric file.




(a) (b)

Figure 2: (a) Spectrally integrated LW flux profile obtained using the nominal absorption optical depth of ECRAD over a standard MLS profile; (b) Difference over spectrally integrated LW fluxes computed using nominal absorption optical thicknesses and the absorption optical thicknesses that have been obtained from tabulated data and nominal values of $x_{H_2O}$

.




(a) (b)

Figure 3: (a) Spectrally integrated SW flux profile obtained using the nominal absorption optical depth of ECRAD over a standard MLS profile; (b) Difference over spectrally integrated SW fluxes computed using nominal absorption optical thicknesses and the absorption optical thicknesses that have been obtained from tabulated data and nominal values of $x_{H_2O}$

The linearity of optical properties with water vapor molar fraction has been validated on the following basis: a radiative transfer code (LW and SW) based on anaytic radiative transfer solutions obtained in a clear-sky situation (without scattering) has been adapted in order to use k-distribution spectral data with a variable quadrature order (such as ECRAD spectral data). Two LW radiative transfer computations have been performed with a modified MLS atmospheric profile: the column containing the highest specific humidity has been identified. The vertical profile of $x_{H_2O}$ in the original MLS profile has then been replaced with the data of this specific atmospheric column for all columns of the LES zone (for z $<$ 4 km) and has not the modified out of this zone. The two radiative transfer computations have then used:

- absorption coefficients obtained from ECRAD for the modified MLS atmospheric profile. Results are presented in figure 2(a) for the LW and figure 3(a) for the SW.
- absorption coefficients that have been interpolated from the local values of $x_{H_2O}$ using the tabulation of optical properties established on the basis of 20 values of $x_{H_2O}$ uniformly spaced (on a logarithmic scale) between $10^{-6}$ and 1. Figure 2(b) displays differences between LW radiative flux obtained using this method and previous (reference) results. Figure 3(b) displays the same differences in the SW domain.

Results being extremely close, the possibility to obtain optical properties of the gas mixture from the tabulated values and the local value of $x_{H_2O}$ is considered as validated.

# 3 File format

The atmospheric file contains the following data, separated by comments:

- The number of atmospheric levels $N_{lev}$ (1 integer)
- The number of atmospheric layers $N_{lay}$ (1 integer)
- Ground temperature (1 real) [K]
- Pressure at each level ($N_{lev}$ reals) [Pa]
- Temperature at each level ($N_{lev}$ reals) [K]
- Altitude at each level ($N_{lev}$ reals) [m]
- The nominal water vapor molar fration in the mixture $x_{H_2O}(nominal)$, for each layer ($N_{lay}$ reals) [mol of $H_2O$ per mol of mixture]
- The number of water vapor molar fraction vertical profiles $x_{H_2O}$ for the tabulation ($N_w$).
- For each one of the $N_w$ vertical profiles of $x_{H_2O}$ in the table: the value of $x_{H_2O}$ for each one of the $N_{lay}$ atmospheric layers. ($N_w$x$N_{lay}reals$) [mol of $H_2O$ per mol of mixture]
- The emissivity of the ground in the LW domain (1 real)
- The emissivity if the ground for the SW domain (1 real)
- The number of spectral intervals $N_{b,LW}$ for the LW domain (1 integer)
- The number of spectral intervals $N_{b,SW}$ for the SW domain (1 integer)
- For each $i_b \in [1, N_{b,LW}]$ LW interval:
  - the lower wavenumber of the interval (1 real) [$\mathrm{cm}^{-1}$]
  - the higher wavenumber of the interval (1 real) [$\mathrm{cm}^{-1}$]
  - the number of quadrature points $N_g(i_b)$ (1 integer)
  - the $N_g(i_b)$ values of the quadrature weight for the interval ($N_g(i_b)$ reals)
- For each $i_b \in [1, N_{b,SW}]$ SW interval:
  - the lower wavenumber of the interval (1 real) [$\mathrm{cm}^{-1}$]
  - the higher wavenumber of the interval (1 real) [$\mathrm{cm}^{-1}$]
  - the number of quadrature points $N_g(i_b)$ (1 integer)
  - the $N_g(i_b)$ values of the quadrature weight for the interval ($N_g(i_b)$ reals)
- For each $i_b \in [1, N_{b,LW}]$ LW interval, and for each $i_g \in [1, N_g(i_b)]$ quadrature points:
  - the **nominal** absorption coefficient of the gas mixture, for each layer ($N_{lay}$ reals) [$\mathrm{m}^{-1}$]

  - the absorption coefficient of the gas mixture **for each tabulation point** $i_w$, for each layer ($N_w$x$N_{lay}$ reals) [$m^{-1}$]

- For each $i_b \in [1, N_{b,SW}]$ SW interval, and for each $i_g \in [1, N_g(i_b)]$ quadrature point:
  - the **nominal** absorption coefficient of the gas mixture, for each layer ($N_{lay}$ reals) [$m^{-1}$]
  - the absorption coefficient of the gas mixture **for each tabulation point** $i_w$, for each layer ($N_w$x$N_{lay}$ reals) [$m^{-1}$]
- For each $i_b \in [1, N_{b,SW}]$ SW interval, and for each $i_g \in [1, N_g(i_b)]$ quadrature point:
  - the **nominal** scattering coefficient of the gas mixture, for each layer ($N_{lay}$ reals) [$m^{-1}$]
  - the scattering coefficient of the gas mixture **for each tabulation point** $i_w$, for each layer ($N_w$x$N_{lay}$ reals) [$m^{-1}$]

A example of (fortran) routine used to read this file is provided below ("input.for" file):

```
      subroutine read_opt_prop_file(infile,Nw,Nlev,Nlay,
     &     Tsurf,Tspace,pressure,
     &     temperature,height,x_h2o_nominal,x_h2o_tab,
     &     lw_emissivity,sw_emissivity,
     &     Nb_lw,nu_lo_lw,nu_hi_lw,Ng_lw,w_lw,
     &     Nb_sw,nu_lo_sw,nu_hi_sw,Ng_sw,w_sw,
     &     ka_lw_nominal,ka_lw_tab,
     &     ka_sw_nominal,ka_sw_tab,
     &     ks_sw_nominal,ks_sw_tab)
      implicit none
      include 'max.inc'
      include 'formats.inc'
c
c     Purpose: to read the optical properties file for the atmosphere
c
c     Input:
c     + infile: file to read
c
c     Output:
c     + Nw: number of tabulation points of water vapor concentration
c     + Nlev: number of atmospheric levels
c     + Nlay: number of atmospheric layers
c     + Tsurf: temperature of the ground [K]
c     + Tspace: temperature of the ground [K]
c     + pressure: pressure at each level [Pa]
c     + temperature: temperature at each level [K]
c     + height: altitude at each level [m]
c     + x_h2o_nominal: molar fraction of H2O at each level for the nominal profile
c     + x_h2o_tab: molar fraction of H2O at each level for each tabulation point
```

```
c     + lw_emissivity: LW emissivity
c     + sw_emissivity: SW emissivity
c     + Nb_lw: number of LW spectral intervals
c     + nu_lo_lw: lower wavenumber limit of each LW interval [cm^-1]
c     + nu_hi_lw: upper wavenumber limit of each LW interval [cm^-1]
c     + Ng_lw: number of quadrature points in each LW interval
c     + w_lw: quadrature weights for each LW interval
c     + Nb_sw: number of SW spectral intervals
c     + nu_lo_sw: lower wavenumber limit of each SW interval [cm^-1]
c     + nu_hi_sw: upper wavenumber limit of each SW interval [cm^-1]
c     + Ng_sw: number of quadrature points in each SW interval
c     + w_sw: quadrature weights for each SW interval
c     + ka_lw_nominal: LW absorption coefficient [m^-1] for the nominal profile
c     + ka_lw_tab: LW absorption coefficient [m^-1] for each tabulation point
c     + ka_sw_nominal: SW absorption coefficient [m^-1] for the nominal profile
c     + ka_sw_tab: SW absorption coefficient [m^-1] for each tabulation point
c     + ks_sw_nominal: SW scattering coefficient [m^-1] for the nominal profile
c     + ks_sw_tab: SW scattering coefficient [m^-1] for each tabulation point
c
c     I/O
      character*(Nchar_mx) infile
      integer Nw,Nlev,Nlay
      double precision pressure(1:Nlev_mx)
      double precision Tsurf,Tspace
      double precision temperature(1:Nlev_mx)
      double precision height(1:Nlev_mx)
      double precision x_h2o_nominal(1:Nlay_mx)
      double precision x_h2o_tab(1:Nlay_mx,1:Nw_mx)
      double precision lw_emissivity,sw_emissivity
      double precision ka_lw_nominal(1:Ngpoints_lw_mx,1:Nlev_mx)
      double precision ka_lw_tab(1:Ngpoints_lw_mx,1:Nlev_mx,1:Nw_mx)
      double precision ka_sw_nominal(1:Ngpoints_sw_mx,1:Nlev_mx)
      double precision ka_sw_tab(1:Ngpoints_sw_mx,1:Nlev_mx,1:Nw_mx)
      double precision ks_sw_nominal(1:Ngpoints_sw_mx,1:Nlev_mx)
      double precision ks_sw_tab(1:Ngpoints_sw_mx,1:Nlev_mx,1:Nw_mx)
      integer Nb_lw
      double precision nu_lo_lw(1:16)
      double precision nu_hi_lw(1:16)
      integer Ng_lw(1:16)
      double precision w_lw(1:16,1:20)
      integer Nb_sw
      double precision nu_lo_sw(1:14)
      double precision nu_hi_sw(1:14)
      integer Ng_sw(1:14)
      double precision w_sw(1:14,1:20)
```

```
c       temp
        logical file_ex
        integer ilay,ilev,ib,ig,i,iw
c       label
        character*(Nchar_mx) label
        label='subroutine read_opt_prop_file'

        inquire(file=trim(infile), exist=file_ex)
        if (file_ex) then
           write(*,*) 'Reading file: ',trim(infile)
        else
           call error(label)
           write(*,*) 'File not found:'
           write(*,*) trim(infile)
           stop
        endif

        open(11,file=trim(infile))
        read(11,*)
        read(11,*) Nlev
        read(11,*)
        read(11,*) Nlay
        read(11,*)
        read(11,*) Tsurf
        Tspace=3.0D+0                   ! K
        read(11,*)
        do ilev=1,Nlev
           read(11,*) pressure(ilev)
        enddo                           ! ilev
        read(11,*)
        do ilev=1,Nlev
           read(11,*) temperature(ilev)
        enddo                           ! ilev
        read(11,*)
        do ilev=1,Nlev
           read(11,*) height(ilev)
        enddo                           ! ilev
        read(11,*)
        do ilay=1,Nlay
           read(11,*) x_h2o_nominal(ilay)
        enddo                           ! ilay
        read(11,*)
        read(11,*) Nw
        do iw=1,Nw
           read(11,*)
```

```
      do ilay=1,Nlay
         read(11,*) x_h2o_tab(ilay,iw)
      enddo                          ! ilay
    enddo                            ! iw
    read(11,*)
    read(11,*) lw_emissivity
    read(11,*)
    read(11,*) sw_emissivity
    read(11,*)
    read(11,*) Nb_lw
    read(11,*)
    read(11,*) Nb_sw
    do ib=1,Nb_lw
      read(11,*)
      read(11,*)
      read(11,*) nu_lo_lw(ib)
      read(11,*)
      read(11,*) nu_hi_lw(ib)
      read(11,*)
      read(11,*) Ng_lw(ib)
      read(11,*)
      do ig=1,Ng_lw(ib)
         read(11,*) w_lw(ib,ig)
      enddo                          ! ig
    enddo                            ! ib
c   SW: per interval
    do ib=1,Nb_sw
      read(11,*)
      read(11,*)
      read(11,*) nu_lo_sw(ib)
      read(11,*)
      read(11,*) nu_hi_sw(ib)
      read(11,*)
      read(11,*) Ng_sw(ib)
      read(11,*)
      do ig=1,Ng_sw(ib)
         read(11,*) w_sw(ib,ig)
      enddo                          ! ig
    enddo                            ! ib
c
c   LW: per interval, per g-point
    i=0
    do ib=1,Nb_lw
      do ig=1,Ng_lw(ib)
         i=i+1
```

```
            read(11,*)
            do ilay=1,Nlay
               read(11,*) ka_lw_nominal(i,ilay)
            enddo                    ! ilay
            do iw=1,Nw
               read(11,*)
               do ilay=1,Nlay
                  read(11,*) ka_lw_tab(i,ilay,iw)
               enddo                 ! ilay
            enddo                    ! iw
         enddo                       ! ig
      enddo                          ! ib
c
c     SW: per interval, per g-point
c     absorption coefficient
      i=0
      do ib=1,Nb_sw
         do ig=1,Ng_sw(ib)
            i=i+1
            read(11,*)
            do ilay=1,Nlay
               read(11,*) ka_sw_nominal(i,ilay)
            enddo                    ! ilay
            do iw=1,Nw
               read(11,*)
               do ilay=1,Nlay
                  read(11,*) ka_sw_tab(i,ilay,iw)
               enddo                 ! ilay
            enddo                    ! iw
         enddo                       ! ig
      enddo                          ! ib
c
c     scattering coefficient
      i=0
      do ib=1,Nb_sw
         do ig=1,Ng_sw(ib)
            i=i+1
            read(11,*)
            do ilay=1,Nlay
               read(11,*) ks_sw_nominal(i,ilay)
            enddo                    ! ilay
            do iw=1,Nw
               read(11,*)
               do ilay=1,Nlay
                  read(11,*) ks_sw_tab(i,ilay,iw)
```

```
            enddo                    ! ilay
          enddo  ! iw
        enddo                        ! ig
      enddo                          ! ib
      close(11)

      return
      end
```